<commit_before>{ buildPythonApplication, fetchFromGitLab, lib
, numpy, h5py, pyqt5, qtpy, future, vtk
} :

buildPythonApplication rec {
  pname = "Pegamoid";
  version = "2.6.2";

  src = fetchFromGitLab {
    owner = "jellby";
    repo = pname;
    rev = "v${version}";
    hash = "sha256-36ZZK2cACvrwn5EVuv+zu6oBsh3vPCZDxQGzKe4PPlg=";
  };


  # The samples and screenshots directories confuse setuptools and they
  # refuse to build as long as these directories are present.
  prePatch = "rm -rf samples screenshots";
  patches = [ ./pipVTK.patch ./Compat.patch ];

  propagatedBuildInputs = [
    numpy
    h5py
    pyqt5
    qtpy
    vtk
    future
  ];

  meta = with lib; {
    description = "Python GUI for OpenMolcas";
    homepage = "https://pypi.org/project/Pegamoid";
    license = licenses.asl20;
    platforms = platforms.linux;
    maintainers = [ maintainers.sheepforce ];
    mainProgram = "pegamoid.py";
  };
}
<commit_msg>pegamoid: fix qt wrapping<commit_after>{ buildPythonApplication, python3, fetchFromGitLab, lib
, numpy, h5py, pyqt5, qtpy, future, vtk, qt5
} :

buildPythonApplication rec {
  pname = "Pegamoid";
  version = "2.6.2";

  src = fetchFromGitLab {
    owner = "jellby";
    repo = pname;
    rev = "v${version}";
    hash = "sha256-36ZZK2cACvrwn5EVuv+zu6oBsh3vPCZDxQGzKe4PPlg=";
  };


  # The samples and screenshots directories confuse setuptools and they
  # refuse to build as long as these directories are present.
  prePatch = "rm -rf samples screenshots";
  patches = [ ./pipVTK.patch ./Compat.patch ];

  preConfigure = ''
    export PYTHONPATH=$PYTHONPATH:${vtk}/${python3.sitePackages}
  '';

  propagatedBuildInputs = [
    numpy
    h5py
    pyqt5
    qtpy
    vtk
    future
  ];

  nativeBuildInputs = [ qt5.wrapQtAppsHook ];

  preFixup = ''
    wrapQtApp "$out/bin/pegamoid.py"
  '';

  meta = with lib; {
    description = "Python GUI for OpenMolcas";
    homepage = "https://pypi.org/project/Pegamoid";
    license = licenses.asl20;
    platforms = platforms.linux;
    maintainers = [ maintainers.sheepforce ];
    mainProgram = "pegamoid.py";
  };
}

{'fraction_non_alphanumeric': 0.11912672045562411, 'fraction_numerical': 0.019933554817275746, 'mean_word_length': 3.199203187250996, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4203379', 'n_tokens_mistral': 812, 'n_tokens_neox': 768, 'n_words': 169}